Lecture 3

Recursion

x The factorial function:

fun fact(n:int) = if n=0 then 1 else n*fact(n-1);
> val fact = fn : int -> int
fact 5;
> val it = 120 : int

x Compiler automatically detects recursive calls

fun f n : int = n+1;
> val f = fn : int -> int
fun f(n:int) = if n=0 then 1 else n*f(n-1);
> val f = fn : int -> int
f 3;
> val it = 6 : int

x f 3 results in the evaluation of 3*f(2) x In old MLs, first f would have been used u f(2) would evaluate to 2+1=3 u hence the expression f 3 would evaluate to 3*3=9

Alternative specification of recursion

x Can use val and fn to define functions

fun f n : int = n+1;
> val f = fn : int -> int
val f = fn (n:int) => if n=0 then 1 else n*f(n-1);
> val f = fn : int -> int
f 3;
> val it = 9 : int

x f in n*f(n-1) is the previous version of f x rec after val forces recursion

fun f n : int = n + 1;
> val f = fn : int -> int
val rec f = fn (n:int) => if n=0 then 1 else n*f(n-1);
> val f = fn : int -> int
f 3;
> val it = 6 : int

Testing equality

x Concrete values like integers, booleans and strings can be tested for equality x Values of simple datatypes, like pairs and records, whose components have concrete types can also be tested for equality

u (v_1,v_2) = (v0_1,v0_2) iff v_1 = v0_1 and v_2 = v0_2

x Many values can be tested for equality u there are infinitely many of them

Equality types

x In general, cannot test equality of functions x Thus not possible to overload = on all types x In old ML = was interpreted on all types u functions were equal if their addresses were equal u if test yielded true then functions equal u but many mathematically (i.e. extensionally) equal functions come out different x Types whose values can be tested for equality are equality types u Equality type variables range over equality types u equality type variables have the form ’’α u ordinary type variables have the form ’α x = has type ’’a * ’’a -> bool

Examples of equality types

x The ML typechecker infers types containing equality type variables

fun Eq x y = (x = y);
> val Eq = fn : ’’a -> ’’a -> bool

fun EqualHd l1 l2 = (hd l1 = hd l2);
> val EqualHd = fn : ’’a list -> ’’a list -> bool

x Trying to instantiate an equality type variable to a functional type results in an error

hd = hd;
> Error: operator and operand don’t agree
> operator domain: ’’Z * ’’Z
> operand: (’Y list -> ’Y) * (’X list -> ’X)

EqualHd [hd] [hd];
> Error: operator and operand don’t agree
> operator domain: ’Z * ’Z
> operand: ’’Y list -> ’’Y list -> bool

x Equality types are controversial:

u benefits not worth the messiness u future versions of ML may omit them

Pattern matching

x Functions can be defined by pattern matching

fun fact 0 = 1
 | fact n = n * (fact(n-1));
> val fact = fn : int -> int

fact 6;
> val it = 720 : int

x Suppose function f is defined by

fun f p_1 = e_1
 | f p_2 = e_2
    .
 | f pn = en

x f e is evaluated by:

u matching e’s value with p_1, p_2, . . ., p_n (that order) u until a match is found, say with p_i u value of f e is then value of e_i

x variables in patterns are locally bound to bits of e they match

Patterns

x Patterns need not be exhaustive

- fun foo 0 = 0;
***Warning: Patterns in Match not exhaustive: 0=>0
> val foo = Fn : int -> int

x What if a function is applied to an argument whose value doesn’t match any pattern?

u a run-time error called an exception results

u exception are covered later

- foo 1;
Exception raised at top level
Exception: Match raised

x The wildcard “ ” matches anything

fun null [] = true
 | null _ = false;
> val null = fn : ’a list -> bool

Examples

x functions hd and tl can be defined by:

fun hd(x::l) = x;
> Warning: match nonexhaustive
> val hd = fn : ’a list -> ’a

fun tl(x::l) = l;
> Warning: match nonexhaustive
> val tl = fn : ’a list -> ’a list

x Almost the same results as the built-in functions

u on [] they give different exceptions

hd []; (* built-in "hd" *)
> uncaught exception Hd

fun hd(x::l) = x;
> Warning: match nonexhaustive
> val hd = fn : ’a list -> ’a

hd[]; (* redefined "hd" *)
> uncaught Match exception std_in:0.0-0.0

The as construct in patterns

x x as p is a pattern that

u matches the same things as p

u binds value matched to x

fun RemoveDuplicates[] = []
 | RemoveDuplicates[x] = [x]
 | RemoveDuplicates(x1::x2::l) =
     if x1=x2 then RemoveDuplicates(x2::l)
              else x1::RemoveDuplicates(x2::l);

>val RemoveDuplicates = fn : ’’a list -> ’’a list

RemoveDuplicates[1,1,1,2,3,4,5,5,5,5,5,6,7,8,8,8];
> val it = [1,2,3,4,5,6,7,8] : int list

x Using as:

fun RemoveDuplicates[] = []
 | RemoveDuplicates(l as [x]) = l
 | RemoveDuplicates(x1::(l as x2::_)) =
     if x1=x2 then RemoveDuplicates l
              else x1::RemoveDuplicates l;

Repeated variables not allowed

x Alas:

fun RemoveDuplicates[] = []
 | RemoveDuplicates(l as [x]) = l
 | RemoveDuplicates(x::(l as x::_)) =
     RemoveDuplicates l
 | RemoveDuplicates(x::l) =
     x::RemoveDuplicates l;
> Error: duplicate variable in pattern(s): x

Anonymous functions can use patterns

x fn-expressions can use patterns

u fn p_1 => e_1 | . . . | p_n => e_n

fn [] => "none"
| [_] => "one"
| [_,_] => "two"
| _ => "many";
> val it = fn : ’a list -> string

(it [], it[true], it[1,2], it[1,2,3]);
> val it = ("none","one","two","many")

Patterns and records

x Patterns can be constructed out of records u “...” (three dots) acts as a wildcard

fun IsMale({sex="male",...}:persondata) = true
 | IsMale _ = false;
> val IsMale = fn : persondata -> bool
IsMale MikeData;
> val it = true : bool

x An alternative definition

fun IsMale({sex=x,...}:persondata) = (x = "male");

x A more compact form of this is allowed

fun IsMale({sex,...}:persondata) = (sex = "male");

x The field name sex doubles as a variable u {· · ·,v,· · ·} abbreviates {· · ·,v=v,· · ·}

The case construct

x The following are equivalent:

u case e of p_1 => e_1 | . . . | p_n => e_n

u (fn p_1 => e_1 | . . . | p_n => e_n) e

fun RemoveDuplicates[] = []
 | RemoveDuplicates(l as [x]) = l
 | RemoveDuplicates(x1::(l as x2::_)) =
     case x1=x2
     of true => RemoveDuplicates l
     | _ => x1::RemoveDuplicates l;
> val RemoveDuplicates = fn : ’’a list -> ’’a list

Exceptions

x Exceptions are a kind of event that occur during evaluation x Can result from run-time errors

u e.g. 1/0 x Can be generated explicitly

u e.g. raise Ex

Exception values

x Exception values are ML values of type exn x exn is a datatype

u datatypes are explained later x Exception value constructors:

u delared using keyword exception u can be used in patterns

Exception packets

x An exception event raises an exception packet

x An exception packet is a raised exception value u Exception packets break normal flow-of-control u they can be trapped using a handler

e handle p_1 => e_1 | . . . | pn => en

x Expression evaluation either:

u succeeds with a value u raises an exception

i.e. fails with an exception packet u doesn’t terminate

Raised exceptions

x Functions can raise exceptions at run-time x A special kind of value is propagated u called an exception packet u usually identifies the cause of the exception x Exception packets have names

u usually reflect the function that raised the exception x Exception packets may also contain values

hd(tl[2]);
> uncaught exception Hd
1 div 0;
> uncaught exception Div
(1 div 0)+1000;
> uncaught exception Div

Declaring exceptions

x Exceptions are declared using the keyword

exception

u they have type exn

x Exceptions are raised by evaluating raise e

u where e evaluates to an exception value

exception Ex1;exception Ex2;
> type exn
  con Ex1 = - : exn
> type exn
  con Ex2 = - : exn

- [Ex1,Ex2];
> [-,-] : exn list

- raise hd it;
Exception raised at top level
Warning: optimisations enabled -
         some functions may be missing from the trace
Exception: Ex1 raised

Exception packets

x “exception name of ty” declares u an exception packet constructor called name u that constructs packets containing values of type ty

exception Ex3 of string;
> exception Ex3
Ex3;
> val it = fn : string -> exn
raise Ex3 "foo";
> uncaught exception Ex3

x The type exn is a datatype

u see later x Exceptions can be used in patterns u useful for handling exceptions

Handling exceptions

x Exceptions are trapped using exception handlers x Example: trapping all exceptions:

u Value of “e_1 handle => e_2” is u value of e_1, unless e_1 raises an exception u in which case it is the value of e_2

hd[1,2,3] handle _ => 0;
> val it = 1 : int
hd[] handle _ => 0;
> val it = 0 : int
hd(tl[2]) handle _ => 0;
> val it = 0 : int
1 div 0 handle _ => 1000;
> val it = 1000 : int

Example: half

x The function half only succeeds on non-zero even numbers

u on 0 it raises Zero

u on odd numbers it raises Odd

exception Zero; exception Odd;
> exception Zero
> exception Odd

fun half n =
 if n=0 then raise Zero
        else let
              val m = n div 2
             in
              if n=2*m then m else raise Odd
             end;
> val half = fn : int -> int

Some examples of using half

half 4;
> val it = 2 : int
half 0;
> uncaught exception Zero
half 3;
> uncaught exception Odd
half 3 handle _ => 1000;
> val it = 1000 : int

x Exceptions can be trapped selectively u by matching the exception packet x If e raises Ex

u value of “e handle Ex_1 => e_1 | . . . | Ex_n => e_n” is u_{the value of ei if Ex equals Exi} u_{otherwise the handle-expression raises Ex}